
MPBASIC

REIHE AUTOMATISIERUNGSTECHNIK Band 216
Programmieren mit BASIC
Siegmar Mdller

1. Auflage
© VEB Verlag Technik, Berlin, 1985

7. Tiny MPBASICS. 60 ff.

7.1. Tiny BASIC

Tiny"” BASIC ist eine aus BASIC abgeleitete, sehr stark vereinfachte Programmiersprache. Die
Vereinfachungen bestehen im wesentlichen in folgenden Punkten:

Es sind nur die Variablen A ... Z zugelassen.
Die einzige Datenart sind ganze Zahlen.

Es gibt keine Felder.

Es gibt keine Strings.

Tiny BASIC hat weniger Anweisungen.

IR

Es fehlen eine Reihe von numerischen Funktionen.

Diese Vereinfachungen gestatten es, Interpreter fur Tiny BASIC mit sehr geringem Speicherbedarf zu
schreiben. Der Speicherplatz fir einen ,normalen, BASIC-Interpreter liegt zwischen 8 bis Uber 20
KByte, wahrend man Interpreter fur Tiny BASIC in 2 ... 4 KByte unterbringt. Die ausschlieBliche
Verwendung ganzzahliger Variablen und Konstanten ergibt dazu eine verhaltnismalig schnelle
Arithmetik. Diese beiden Vorteile stehen dem Nachteil des geringeren Komforts gegentber. Damit
ergeben sich fur Tiny BASIC andere Anwendungsbereiche als fur ein ,grofSes” BASIC. Man setzt diese
Sprache in sehr kleinen Rechnern mit wenig Speicherplatz zum Steuern und Regeln ein. Dabei ist man
durchaus in der Lage, auch Echtzeit Probleme zu I6sen.

Ein Echtzeitproblem besteht dann, wenn der Rechner auf ein externes Ereignis innerhalb einer
vorgegebenen Zeit in einer diesem Ereignis entsprechenden Art und Weise reagieren muss. Ob eine
Sprache mit einem gegebenen Rechner echtzeitfahig ist, hangt also auch von dem zu l6senden
Problem ab. Wenn es um Zehntelsekunden geht, dann ist Tiny BASIC oft schon einsetzbar, wahrend
man im Mikrosekundenbereich nur mit Maschinenprogrammen Herr der Lage bleiben kann.

Ein generelles Problem bei der Anwendung héherer Programmiersprachen in Echtzeitproblemen
besteht darin, dass es nicht moglich ist, wie bei der Assemblerprogrammierung, die Abarbeitungszeit
exakt zu berechnen, In der Praxis gentgt es aber fast immer, diese Zeit zu messen. Dabei sollte man
jedoch genau Uberlegen, mit welchen Eingangsdaten man bei dem zu testenden Programmteil den
ungunstigsten Fall erzeugt. Dadurch wird sichergestellt, dass sich nicht bei anderen Daten eine
langere Abarbeitungszeit ergibt. Zum Messen geniugt meist schon die Armbanduhr, wenn man das zu
testende Programmestuck in eine Schleife fasst. Zum Beispiel:

10 FOR I=1 TO 1000

programmstick
1000 NEXT I

Misst man hier x Sekunden, dann dauert die Abarbeitung des Programmstlcks etwas weniger als x
Millisekunden. (Weniger, weil FOR und NEXT noch hinzu kommen)

In der Praxis sind jedoch meist nur wenige Programmstellen zeitkritisch, und Tiny BASIC lasst hier
noch die Méglichkeit offen, auf die Maschinenebene auszuweichen (CALL, Interrupts im Hintergrund),
wenn das eben Vorgeschlagene nicht zu befriedigenden Ergebnissen fuhrt. Hat man nach
eingehender Prufung entschieden, Tiny BASIC einzusetzen, dann wird man wesentlich schneller zu
einem fertigen Programm kommen als mit dem Assembler.

7.2. Konzept von Tiny MPBASIC

Wie bei BASIC gibt es auch bei Tiny BASIC eine kaum Uberschaubare Vielzahl von Versionen. Tiny
MPBASIC (winziges Mikroprozessor-BASIC) wurde ursprunglich fur den Einchipmikrorechner U883
geschrieben, kann aber prinzipiell auch fur jeden anderen Prozessor implementiert werden. Zum
Verstandnis der folgenden Ausfihrungen sind dennoch einige Grundkenntnisse tber den Aufbau und
die Programmierung des Einchipmikrorechners U881 erforderlich. Hierzu sei z. B. auf [11] verwiesen.
Wegen des begrenzten Programmspeichers (2 KByte) wurde im U883 lediglich ein
Interpreterprogramm untergebracht, das ein fertiges, sowohl syntaktisch als auch semantisch
fehlerfreies Programm abarbeiten kann. Es fehlen Kommandos zum Editieren und Testen. Sie werden
von einem externen Programm mit der Bezeichnung ,Editor/Debugger,, realisiert. Es kann in dem. mit
dem U883 realisierten System implementiert sein, oder auf einem Wirtsrechner laufen, Diese
Trennung ermdglichte die Schaffung eines verhaltnismaRig leistungsfahigen Interpreters, wahrend der
Editor/Debugger in fertigen Geraten (mit fertig entwickeltem und getestetem Programm) entfallen
kann.

Der Interpreter gestattet die Einbindung von Programmen in Maschinensprache. Wir sprechen hierbei
von Prozeduren. Eine Prozedur in Tiny MPBASIC ist ein in Maschinensprache geschriebenes
Programm, das Daten vom Interpreter Ubernimmt, verarbeitet, Daten an ihn zurickgibt und mit
einem Namen aufgerufen werden kann. Eine Funktion ist demzufolge eine Prozedur, die genau einen
Wert an den Interpreter zurtckgibt. Die Datenlbergabe wird vom Interpreter unterstitzt. Die
Zuordnung des Prozedurnamens zur Startadresse des Maschinenprogramms erfolgt vermittels einer
Tabelle, der Prozedurtabelle. Sie ist vom Nutzer zu erstellen. Dadurch kdnnen mit diesem Tiny BASIC,
trotz Einbeziehens der Maschinenebene, maschinenunabhangige Programme verfasst werden. Beim
Ubergang von einem Rechner auf einen anderen genligt es, die benutzten Prozeduren neu oder
umzuschreiben, wahrend das BASIC- Programm unverandert ubernommen werden kann. Neben der
Maglichkeit, selbst Prozeduren (und Funktionen) zu erstellen, bietet der Interpreter dem Anwender
eine Reihe bereits fest vorgegebener Prozeduren und Funktionen.

Tiny MPBASIC gestattet daruber hinaus die Verarbeitung von Traps (Fallen). Das sind
Programmunterbrechungen, die beim Erfulltsein einer gewissen Bedingung vom Interpreter
softwaremafig ausgelést werden. Die Bedingung wird vom Programmierer festgelegt (s. Abschn.
7.3.2.).

Ein weiteres Problem bei der Schaffung eines moglichst gerateunabhangigen Interpreters sind die
Eingabe- und die Ausgabeschnittstellen fur INPUT bzw. PRINT. Um hier alle MAglichkeiten offen zu
lassen, ruft PRINT ein Unterprogramm PUTCHR auf, das ein Zeichen ausgibt und vom Anwender des
U883 erstellt wird. Wenn PRINT eine Zeile ausdruckt, dann wird PUTCHR so oft aufgerufen, wie

Zeichen in dieser Zeile sind. Zum Einlesen von Zahlen ruft INPUT ein, ebenfalls vom Anwender
geschriebenes, Programm GETCHR auf, das bei jedem Aufruf ein Zeichen einliest.

7.3. Interpreter

7.3.1. Ausdriicke

Wir unterscheiden in Tiny MPBASIC arithmetische und logische Ausdriicke. Letztere haben immer
die Gestalt

aausdruck vop aausdruck ,

wobei aausdruck ein arithmetischer Ausdruck und vop einer der Vergleichsoperatoren

>, <,>=, <=, = und < >

ist. Das Ergebnis eines logischen Ausdrucks ist eine logische GroRe (wahr oder falsch) und kein
Zahlenwert. Hier besteht ein grundsatzlicher Unterschied zu anderen BASIC- Interpretern, der
allerdings fur die Praxis untergeordnete Bedeutung hat. Logische Ausdrucke dirfen nurin
Anweisungen stehen, die logische GroRen verarbeiten konnen.

In arithmetischen Ausdrucken werden ganzzahlige Konstanten, Variablen und Funktionen
verknUpft. Da es GleitkommagroRen ohnehin nicht gibt, entfallt die Kennzeichnung mit %. Dieses
Zeichen hat in Tiny MPBASIC eine andere Bedeutung. Vor einer Konstanten kennzeichnet es die
hexadezimale Darstellung. Der zuldssige Bereich liegt zwischen %0000 und %FFFF bei hexadezimaler
und zwischen -32767 und 32767 bei dezimaler Schreibweise. (Die Zahl -32768 kann intern
verarbeitet, aber nicht ein- bzw. ausgegeben werden.) Gerechnet wird im Zweierkomplement.
Naheres zu dieser Darstellung und dem Rechnen damit findet der Leser z. B. in [9].

Es gibt die arithmetischen Operatoren +, -, *, / und $MOD.
$MOD, modulo, liefert den Rest, den eine ganze Zahl bei der Division durch eine andere ganze Zahl
gibt. Es qilt

| A$MOD B | = | A - (A/B*B) |

Bei der Division wird der gebrochene Teil weggelassen. Das Dollarzeichen vor MOD hat nichts mit
einem String zu tun; Strings sind nicht implementiert. Es kennzeichnet vielmehr $MOD als einen
Operator. Das trifft auch auf die bitweisen logischen Operatoren

$AND, $OR und $XOR

zu. Diese logischen logischen Operatoren fuhren fur jedes Paar von gleichwertigen Binarstellen ihrer
Operanden in der internen Zweierkomplementdarstellung die betreffende logische Operation aus. lhre
Benutzung setzt voraus, dass Zweierkomplement, binare Darstellung und hexadezimale Darstellung
fur den Programmierer vertraute Begriffe sind. Beispielsweise kann man $AND zum ,Maskieren” von
Bits in einem Datenwort benutzen. Mit

A $AND 1

Homecomputer DDR - https://hc-ddr.hucki.net/wiki/



Last update: 2024/02/02 13:47 elektronik:u883:mpbasic https://hc-ddr.hucki.net/wiki/doku.php/elektronik/u883/mpbasic

erhalt man den Zustand vom Bit 0 in A, indem mit der UND-Verkndpfung die restlichen Bits
zuruckgesetzt werden. Mit $OR kdnnen Bits auf Eins gesetzt werden.
$XOR eignet sich u. a. zum bitweisen Vergleich von Daten. Der Ausdruck

A $XOR %70 $AND %FO

liefert z. B. genau dann den Wert Null, wenn in A die Bits 4, 5 und 6 Eins sind und Bit 7 Null ist. Alle
bitweisen logischen und arithmetischen Operatoren haben dieselbe Prioritat. Sie dirfen gemeinsam in
arithmetischen Ausdricken auftreten. Damit werden arithmetische Ausdricke streng von links
nach rechts abgearbeitet. Wird eine andere Reihenfolge der Berechnung gewunscht, so ist das mit
Klammern zu regeln. Weiterhin stehen die in Tafel 8 aufgefuhrten allgemeinen, d. h.
maschinenunabhangigen Funktionen zur Verfligung.

Die Funktionsparameter werden grundsatzlich in eckige Klammern geschrieben.

Beim Rotieren nach links werden alle 16 Bits um eine Stelle nach links geschoben, und das
hochstwertige Bit wird zum niederwertigsten Bit. Rotiert man z. B. die Zahl -32767, die intern die
binare Darstellung 1000 0000 0000 0001 hat, nach links, so ergibt das 0000 0000 0000 0011 (binar)
bzw. 3 in dezimaler Schreibweise. Das Rotieren nach rechts funktioniert prinzipiell genauso, nur eben
in die andere Richtung. RR ist somit die inverse Funktion zu RL. Programmiert man RR[RL[x]], so
ergibt das immer x. Man braucht diese Funktion z. B., wenn Eingabeports multiplex abgefragt werden
sollen.

Tafel 8, Funktionen in Tiny MPBASIC

ABS[x] absoluter Betrag von x

NOT[x] bitweise logische Negation von x

GTC ein Zeichen von der Konsole holen (benutzt GETCHR)
INPUT  Zahl von der Konsole holen

RL[x] X um ein Bit nach links rotieren

RR[x] X um ein Bit nach rechts rotieren

Tafel 9. Maschinenorientierte Funktionen im U8S3

GETR[r] Inhalt des Registers r lesen

GETRRI[r] Inhalt des Registerpaares r, r+l1 lesen

GETEB[a] Inhalt des auf der Adresse a abgespeicherten Bytes im
Datenspeicher lesen

GETEW[a] Inhalt der auf den Adressen a, a+l abgespeicherten Bytes im
Daten-Speicher lesen

Daruber hinaus gibt es fur den U883 einige Funktionen, mit denen man auf Register bzw. auf den
Datenspeicher zugreifen kann (Tafel 9). Dadurch hat der Anwender sofort, d. h. ohne erst eigene
Funktionen schreiben zu mussen, Zugriff auf fast alle Schaltkreisfunktionen. BASIC-Programme, die
von diesen Funktionen Gebrauch machen, sind naturlich nicht unverandert auf einen anderen
Rechner Ubertragbar. Deshalb sind diese vier Funktionen nicht Bestandteil der Sprache Tiny MPBASIC.
Wirde man etwa einen Interpreter fur den U880 schreiben, so waren hier die Funktionen

GETB[a] Byte aus dem Speicher von Adresse a lesen
GETW[a] Bytes aus dem Speicher von den Adressen a und a+1l lesen
IN[p] Port p lesen

sinnvoll.

7.3.2. Anweisungen

Da uns die BASIC-Anweisungen i. allg. schon bekannt sind, genugt es, hier die davon in Tiny MPBASIC
implementierten mit der Syntaxvorschrift aufzufihren, und nur die Besonderheiten zu erlautern (Tafel
10). Eine Anweisung besteht aus ihrem Namen, z. B. LET, und den Argumenten, z. B. A=0. Soll eine
Anweisung mehrmals nacheinander ausgefuhrt werden, so genugt es, den Namen nur einmal und
lediglich die Argumente mehrfach aufzuschreiben, wenn man letztere mit Komma voneinander trennt.
Zum Beispiel

10 LET A=0, B=0, I=10

Tafel 10. Anweisungen in Tiny MPBASIC
LET variable = aausdruck

GOTO aausdruck

IF lausdruck THEN anweisung

IF lausdruck THEN anweisungsblock
ELSE anweisung

ELSE anweisungsblock

GOSUB aausdruck

RETURN

INPUT {"text"} variable

PRINT {"text"} {aausdruck}

PRINTHEX {"text"} {aausdruck}

STOP

END

REM {kommentar}

CALL aausdruck

PROC {variablenliste =} prozedurname {parameterliste}
TRAP lausdruck TO aausdruck

CLRTRP

WAIT aausdruck

Bei der Anweisung GOTO ist ein Ausdruck zur Angabe der Zeilennummer zugelassen. Gleiches trifft
auf GOSUB zu. Die Anweisungen ON...GOTO und ON...GOSUB sind dadurch leicht zu umgehen. Ein
Anweisungsblock besteht aus beliebig vielen, durch Semikolon getrennten Anweisungen, die auf einer
Zeile stehen. Es ist mdglich, mehrere Anweisungen auf eine Zeile zu schreiben, wenn man diese
durch Semikolon voneinander trennt. Bei dem in Anfuhrungsstriche gesetzten Text in den
Anweisungen INPUT, PRINT und PRINTHEX handelt es sich im Grunde um eine Stringkonstante. Strings
gibt es ansonsten nicht. INPUT druckt zunachst den programmierten Text aus und erwartet dann die
Eingabe einer Zahl (dezimal oder hexadezimal). Wie wir bereits sahen (Tafel 8), gibt es INPUT auch
als Funktion. Dort kann kein Text ausgedruckt werden. Es erscheint statt dessen ein Fragezeichen.
INPUT hat als Funktion den Vorteil, dass es wie jede Funktion in Ausdrucken benutzt werden kann. Will
man z. B. den Nutzer seines Programms eine Temperatur in Zehntel Grad Celsius eingeben lassen, die
jedoch eigentlich in Zehntel Grad Kelvin bendtigt wird, so kann man einfach

10 LET T = INPUT+2732

programmieren.

Die Anweisung PRINTHEX unterscheidet sich von PRINT nur dadurch, dass das Ergebnis des

Ausdrucks hexadezimal ausgegeben wird. Will man mehrere Ausgaben auf eine Zeile bringen, so
benutzt man bei Tiny MPBASIC anstelle des Semikolons (dieses wird schon zum Trennen von
Anweisungen benutzt) das Komma. Das Drucken in Spalten ist nicht maglich.

Mit der Anweisung PROC werden Prozeduren vom BASIC-Interpreter aufgerufen. Die Variablenliste
sind in eckige Klammern gesetzte und durch Komma getrennte Variablen. Bei der Parameterliste
handelt es sich um in eckige Klammern gesetzte und durch Komma getrennte Ausdricke. Zum
Beispiel

10 PROC [A, B, X] = MPROG [3*A, X, 2]

Es gibt bereits implementierte Prozeduren (Tafel 11). Der Sinn der Prozeduren besteht allerdings vor
allem darin, dass der Anwender, wenn er selbst welche schreibt, Tiny MPBASIC einem gegebenen
Anwendungsfall optimal anpassen kann. Mit Prozeduren realisiert man elementare Steuerfunktionen,
die in BASIC nicht oder nur umstandlich zu programmieren waren, oder deren Realisierung in BASIC
ein zu langsames Programm ergeben wurde.

Tafel 11. Prozeduren im U883

PTC[z] Zeichen z an die Konsole ausgeben (benutzt PUTCHR)

SETR[r,w] Register r mit dem Wert w belegen

SETRR[r,w] Registerpaar r, r+l mit dem Wert w belegen

SETEB[a,w] Byte im Datenspeicher auf der Adresse a mit dem Wert w belegen
SETEW[a,w] Bytes im Datenspeicher auf den Adressen a, a+l mit dem Wert w
belegen

Die SET-Prozeduren in Tafel 11 sind, wie die GET-Funktionen, wiederum nicht Bestandteil der Sprache
Tiny MPBASIC. Sie haben ebenfalls nur den Zweck, den Interpreter im U883 fur den Anwender sofort,
d. h. ohne daR er selbst erst eine Prozedur schreiben muss, nutzbar zu machen.

Eine weitere, speziell fur Steuerungen gedachte Anweisung, ist TRAP (Falle). Nachdem TRAP
abgearbeitet wurde, testet der Interpreter vor der Abarbeitung jeder neuen Programmzeile die
gesetzte Trapbedingung (lausdruck). Sobald diese erflllt ist, erfolgt automatisch ein GOSUB zu der
durch aausdruck gegebenen Zeile. Die Bedingung wird vorher geloscht. Danach wird die Traproutine
abgearbeitet, bei der es sich um ein gewdhnliches Unterprogramm handelt, welches mit RETURN
endet. RETURN bringt die Programmabarbeitung schlief8lich wieder zu der Stelle, an der sie
unterbrochen wurde. Soll die Trapbedingung danach weiter Uberwacht werden, so muss in der
Traproutine eine entsprechende TRAP-Anweisung gestanden haben. Es kann immer nur eine
Bedingung aktiv sein. Das Setzen einer neuen fuhrt zum Loschen der alten Bedingung. Oft ist es
jedoch maglich, eine Bedingung so zu formulieren, dass sie mehrere in sich vereint, wenn das notig
sein sollte. Welche dann davon das Trap ausgel6st hat, kann in der Traproutine ermittelt werden. An
einem Anwendersystem mogen z. B. zwei 8-Bit-A/D-Wandler angeschlossen sein, die mit den selbst
geschriebenen Funktionen AD1 und AD2 abgefragt werden kdnnen. Wenn AD1>200 oder AD2>180
wird, soll ein Trap ausgelost werden. Dann kann man

10 TRAP AD1/201 $O0R (AD2/181)>0 TO ...

programmieren. Die Division AD1/201 ergibt so lange Null, wie AD1«200 ist, weil die Stellen nach
dem Komma abgeschnitten werden. Genauso bleibt AD2/181 Null, wenn AD2«180 ist. Uberschreitet
einer der beiden, AD1 oder AD2, den vorgeschriebenen Maximalwert, so liefert die Verknupfung $OR
ein Ergebnis >0, und es kommt zum Trap.

Will man die Trapbedingung nur I6schen, ohne dabei eine neue zu setzen, so gibt man die Anweisung
CLRTRP (clear trap, l6sche Trap).

SchlieBlich sei noch WAIT erlautert. Diese Anweisung berechnet zunachst den Ausdruck und ruft
danach eine Software-Warteschleife auf, die so oft durchlaufen wird, wie das Ergebnis des Ausdrucks
angibt. Ein Durchlauf dauert genau eine Millisekunde. (Beim U883 wird dabei ein 8-MHz-Quarz
vorausgesetzt.) Bei der Anwendung von WAIT beachte man, dass die Berechnung des Ausdrucks und
die vor- und nachher abzuarbeitenden Anweisungen ebenfalls Zeit brauchen. Weiterhin hat man bei
WAIT mit groRBen Zeiten zu beachten, dass wahrend des Wartens keine Trapbedingung getestet wird.

7.4. Anwendungsbeispiele

Beispiel 1. Trotz des bescheidenen Zahlenbereiches lassen sich eine Reihe von mathematischen
Berechnungen auch in Tiny BASIC mit hinreichender Genauigkeit ausfuhren, wiewohl etwas mehr
Sorgfalt bei der Programmierung erforderlich ist, um Bereichsuberschreitungen zu vermeiden. Wir
wollen mit dem Newton-Verfahren die Quadratwurzel aus einer Zahl berechnen. Dazu wird von einer
Anfangsnaherung X0 ausgegangen und so lange eine bessere Naherung berechnet, bis sich die
erreichbare Genauigkeit eingestellt hat. Die Formel zur Berechnung der verbesserten Naherung Xn+1
aus der Naherung Xn lautet

Xn+1l =Xn-(Xn"2-Y)/(2Xn),

wobei Y die zu radizierende Zahl ist. Wirde man nun
LET X = X-(X*X-Y)/(2*X)

programmieren, so brachte das keine brauchbaren Ergebnisse. Das liegt zum einen daran, dass bei
der Berechnung von xn”2 es bereits fir Xn>181 zu Bereichsuberschreitungen kommt, und zum
anderen wird, was viel schlimmer ist, mit wachsendem n die Differenz zwischen Xn”2 und > immer
kleiner und die Division durch 2*Xn deshalb und wegen der Verwendung von ganzen Zahlen viel zu
ungenau. Das ist jedoch noch kein Grund, das Newton-Verfahren hier zu verwerfen. Statt dessen
Uberlegt man, wie die Formel eventuell umzustellen ist, um die eben festgestellten Schwierigkeiten zu
umgehen. Und tatsachlich ist das moglich. Es ist namlich

Xn+l=Xn-(Xn"2-Y)/(2Xn) =Xn-Xn/2 + Y/ (2Xn)

Bei dieser Umformung sind sowohl das Quadrat als auch die kritische Differenz verschwunden. Wir
kénnen folgendes Programm aufschreiben:

10 INPUT "Y = " Y

20 LET X=1, N=10

30 LET X=X-(X/2)+(Y/X/2)

40 LET N=N-1; IF N>0 THEN GOTO 30
50 PRINT "SQR(Y) =" X

Als Anfangsnaherung benutzen wir die 1, und die zehnte Naherung ist bei unserem Zahlenbereich
immer genau genug.

Beispiel 2. In dem nun folgenden Beispiel benutzen wir den Operator $MOD. Es handelt sich um eine
Primfaktorzerlegung. Man bendtigt so etwas z.B., um die Frequenzen zu ermitteln, die sich leicht aus

einer gegebenen (Quarz-) Oszillatorfrequenz mittels Teilerstufen erzeugen lassen. Das benutzte
Verfahren ist recht einfach. Es werden alle maglichen Faktoren ausprobiert, bis die Zahl schlieBlich
vollstandig zerlegt ist.

10 INPUT "ZAHL: " Z

20 LET F=2

30 IF Z $MOD F=0 THEN LET Z=Z/F; PRINT F
40 ELSE LET F=F+1

50 IF F*F<=Z THEN GOTO 30

60 PRINT Z

Die Zeile 30 lautet verbal umschrieben: Wenn Z durch den Faktor F teilbar ist, dann flhre die Division
aus und drucke F als ermittelten Primfaktor.

Beispiel 3. Wir wollen den Inhalt des Datenspeichers des U883 in einem gewissen Adressbereich
ausdrucken. Dabei soll das folgende Druckbild auf jeder Zeile entstehen:

adresse datenl daten2 ... daten8 ascii

Die Daten geben wir zu je zwei Bytes aus. also kommen 16 Bytes auf jede Zeile. Rechts erscheinen
die Bytes, die als druckbare ASCII-Zeichen interpretiert werden kénnen, als solche ausgedruckt. Damit
kann man leicht abgespeicherte Texte erkennen. Das Programm hierfur sieht folgendermafien aus:

10 REM RAM DUMP

20 INPUT "ADRESSE: " A, "LAENGE: " I

25 IF I<=0 THEN END

27 REM AUF VIELFACHES VON 16 AUFRUNDEN
30 IF I $MOD 16<>0 THEN LET I=I/16+I*16
40 REM EINE ZEILE

45  REM ADRESSE

50 LET J=0; PRINTHEX A, " ",

55 REM DATEN

60 PRINTHEX " " GETEW[A],
70 LET I=I-2, J=J+2, A=A+2
80 IF J <16 THEN GOTO 60

90 LET A=A-16

95 REM ASCII

100 PRINT " ",

110 LET C=GETEB[A]

120 IF C>%1F THEN IF C<%7F THEN PROC PTC[C]
130  ELSE PROC PTC[%2E]; REM PUNKT

140 LET J=J-1, A=A+l

150 IF J>0 THEN GOTO 110

160 PRINT

170 IF I>0 THEN GOTO 50

In Zeile 30 wird die eingegebene Lange auf eine durch 16 teilbare Zahl aufgerundet, damit auch die
letzte Zeile des RAM-Ausdrucks immer voll wird. Zeile 120 druckt alle Zeichen, deren Kode zwischen
%1F und %7F liegt. Handelt es sich jedoch um kein druckbares Zeichen, so gibt Zeile 130 einen Punkt
aus.

Beispiel 4. Wir wollen nun die Benutzung von TRAP mit einem formalen Beispiel illustrieren. Es bringt
daruber hinaus einige typische Details bei der Anwendung von Tiny MPBASIC.

10 REM ZEICHENSATZ AUSDRUCKEN UND

20 REM IM HINTERGRUND TASTENEINGABEN LESEN
30  LET C=%20, A=%1300, I=0

40  GOSUB 250

50 PROC PTCI[C]

60 LET W=100

70 LET W=W-1

80 IF W>0 THEN GOTO 70
90 LET C=C+1

100 IF C<%5F THEN GOTO 50

110 IF I=0 THEN GOTO 30

115 REM EINGABEN WIEDER AUSDRUCKEN
120 LET J=0

130 PROC PTC[%0D]

140  PROC PTC[GETEB[A+J]]

150  WAIT 100; LET J=J+1

160 IF I>J THEN GOTO 140

170 END

200 REM TRAPROUTINE

210 PROC SETR[%FA, GETR[%FA]$AND %F7]
220 PROC SETEB[A+I, GETR[%FO]]

230 LET I=I+1

240 IF I>%FF THEN LET I=0

250 TRAP GETR[%FA]$MOD 8<>0 TO 210
260 RETURN

Dieses Programm druckt fortwahrend alle ASCII-Zeichen mit den Kodes %20 . .. %5E bis wahrend des
Druckens uber die SIO Zeichen empfangen worden sind. Dann werden die empfangenen Zeichen
ausgedruckt, und die Abarbeitung ist beendet. TRAP Uberwacht im Interrupt-Requestregister des
U883 (Reg. %FA) das SIO-Empfangerbit, das in der Trapbedingung mit $AND 8 maskiert wird. Da die
Trapbedingung am Ende der Traproutine ohnehin neu gesetzt werden muss, erfolgt das erste Setzen
mit GOSUB 250 in Zeile 40. Das spart Speicherplatz, Mit LET A=%1300 in Zeile 30 wird die RAM-
Adresse festgelegt, wo die empfangenen Zeichen abgespeichert werden sollen. Die Zeilen 60... 80
bilden eine Warteschleife, die zwischen dem Ausdrucken zweier Zeichen eine Pause erzeugt. Hier
ware es nicht gunstig, WAIT einzusetzen, weil dann so lange die Trapbedingung nicht Uberwacht
werden wurde. In der Traproutine wird zuerst in Zeile 210 das SIO-Bit zurickgesetzt. Das geschieht
dadurch, dass zuerst das gesamte Register gelesen, anschlieend das Bit mittels $AND %F7
ruckgesetzt und schlielBlich das Ergebnis mit SETR wieder in das Register eingeschrieben wird.
Dadurch bleiben die anderen sieben Bits unverandert. In Zeile 220 wird das empfangene Zeichen
gelesen (SIO-Register %F0) und abgespeichert. Wir sehen hier, wie man in Tiny MPBASIC mit einem
eindimensionalen Feld arbeiten kann, obwohl Felder gar nicht implementiert sind. Statt DIM A... zu
programmieren, legt man mit LET A=%1300 eine Adresse fur dieses Feld fest und kann dann mit
SETEB[A+I, ...] oder GETEB[A+1] anstelle von LET A(l) = ... bzw. ... A (l) ... auf dieses Feld zugreifen.
Selbstverstandlich hat man dabei darauf zu achten, dass der Index | keine unzulassigen Werte
annimmt (Zeile 240).

7.5. Benutzen des Interpreters im U883

Der Interpreter im U883 ist als Unterprogramm ausgeflhrt, das ein fertiges, fehlerfreies BASIC-
Programm abarbeiten kann. Der Eintrittspunkt ist %7FD, Vor dem Aufruf ist in die Register 6 und 7 die
Startadresse des BASIC-Programms und in die Register 8 und 9 die Adresse der Prozedurtabelle,
deren Aufbau wir im folgenden mit beschreiben werden, zu laden. Ist eine Prozedurtabelle nicht
vorhanden, dann mussen die Register 8 und 9 gel6scht werden. Der Registerzeiger ist auf %10 zu
setzen.

Das BASIC-Programm wird ohne Leerzeichen im ASCII-Kode abgelegt. Die Zeilennummern werden
binar abgespeichert, wobei jedoch das hochstwertige Bit immer gesetzt ist. Die Namen der
Anweisungen und die Operatoren mit $ werden, wie in Tafel 12 aufgeflhrt, abgekirzt. Als Markierung
fur das Zeilenende dient Carriage Return (%0D), fur das Programmende %00.

Tafel 12. Abkilrzungen in Tiny MPBASIC

LET L

GOTO G

IF ... THEN F ... ;
ELSE >;
GOSUB S
RETURN R
PROC 0
TRAP ... TO ! ... ,
CLRTRP /
INPUT I
PRINT P
PRINTHEX H
STOP T

END E

WAIT W
CALL C

REM M

SAND $A
SOR $0

SXOR $X
SMOD $M

Das folgende kleine Programm, das 1 KByte Speicher ab Adresse %1000 I6scht, wird so, wie in Tafel
13 aufgezeigt, abgespeichert. Zum Ausdrucken der internen Darstellung kann z.B. das im letzten

Abschnitt (Beispiel 2) aufgefuhrte Programm benutzt werden.

10 LET A=%1000, L=%400
20 PROC SETEB[A,0]

30 LET A=A+1, L=L-1
40 IF L>0 THEN GOTO 20

Tafel 13. Interne Darstellung eines Programms in Tiny MPBASIC

2000 80 OA 4C 41 3D 25 31 30 30 30 2C 4C 3D 25 34 30 ..LA=%1000,L=%40

2010 30 OD 80 14 4F 53 45 54 45 42 5B 41 2C 30 5D 0D 0...0SETEB[A,0].
2020 80 1E 4C 41 3D 41 2B 31 2C 4C 3D 4C 2D 31 0D 80 ..LA=A+1,L=L-1..
2030 28 46 4C 3E 30 3B 47 32 30 0D 00 00 00 00 60 00 (FL>0;G20.......

Es sei angenommen, dass dieses Programm auf der Adresse %2000 stehe. Dann kann es mit den
folgenden Maschinenbefehlen aufgerufen werden:

LD 6, #%20
CLR 7

CLR 8

CLR 9

SRP #%10
CALL %07FD

Far die Programme PUTCHR und GETCHR, die das Benutzen von PRINT, PRINTHEX und INPUT
ermdglichen, gelten nachstehende Konventionen: GETCHR hat die Eintrittsadresse %0815, PUTCHR
%0818. Das eingelesene Zeichen (GETCHR) wird im Arbeitsregister R3, das auszugebende Zeichen
(PUTCHR) im Arbeitsregister R5 Ubergeben. Der Registerzeiger steht auf %10. Zur freien Verfigung
hat der Anwender die Register ab %54.

Die Prozedurtabelle, die dem Interpreter die Zuordnung vom Prozedurnamen zur Eintrittsadresse
ermoglicht, hat folgenden Aufbau:

Byte 0|1|2|3|4| . . . |[N|[N+1|N+2|

Byte 0 enthalt die Lange N des Prozedurnamens, die binar abgespeichert wird. (0<N<%FE = 254). Die
Bytes 1 ... N enthalten den Namen im ASCIl-Kode, und die Bytes N+1 und N+2 enthalten die
Eintrittsadresse der Prozedur. Danach darf, beginnend mit der Lange ihres Namens, die nachste
Prozedur in die Tabelle eingetragen werden. Das Ende wird mit %FF gekennzeichnet. Beginnen z. B.
die Prozeduren PROZ und FUNKT auf den Adressen %2010 bzw. %3000, dann musste eine
Prozedurtabelle damit so aussehen:

%04 | %50 | %52 | %4F | %5A | %20 | %10 | %05 | %46 | 555 | S4E | %4B | %54 | %30 | %00 | SFF

Die Ubergabe der Parameter vom Interpreter an die Prozedur und der Ergebnisse von der Prozedur
an den Interpreter erfolgt Uber den Stack. Wenn die Prozedur m Ergebnisse an den Interpreter
ubergibt und n Parameter von diesem Ubernimmt, dann befindet sich der Stack beim Aufruf der
Prozedur in dem in Tafel 14 aufgezeigten Zustand. Vor dem letzten RET in der Prozedur, das wieder
zum Interpreter fuhrt, muss der Stackpointer um 2n-2 erh6ht worden sein (wenn n>1) und naturlich
auf die Ruckkehradresse zum Interpreter zeigen.

Tafel 14. Datenlbergabe bei Prozeduren im U883

Stack Inhalt
SP+2n+2m+1 Platz fur das m-1lte bis erste Ergebnis,
. SP+2n+2 wenn m>1. Der Wert fir das letzte (mte) Ergebnis ist in das
Arbeitsdoppelregister RR2 zu schreiben.
SP+2n+1 ... SP+2n  Wird vom Interpreter benutzt.
SP+2n-1 Erster bis n-1lter Parameter, wenn n>1.

. SP+2 Der Wert des letzten Parameters befindet sich im
Arbeitsdoppelregister RR4.
SP, SP+1 Rickkehradresse zum Interpreter

SP Inhalt vom Stackpointer

Schliel8lich sei noch erwahnt, dal der U883 auf der Adresse %812 startet, falls dort ROM liegt und
sonst auf %E000.

7.6. Editor/Debugger

Vom Hersteller des U883 wird auch der zugehdrige Editor/Debugger, den wir der Kirze halber nur
»Editor” nennen wollen, bereitgestellt. Er meldet sich mit # als Promptzeichen. Danach gibt man als
erstes Kommando NEW, falls noch kein Programm im Speicher stand. Das ist notwendig, weil beim
Einschalten der Speicher nicht automatisch geldscht wird, um ein evtl. in einem CMOS-RAM
abgelegtes Programm zu erhalten. Danach kdnnen Programme eingegeben werden, so wie wir das im
Abschn. 5. kennengelernt haben. Der Editor beseitigt dabei die Leerzeichen, ersetzt die
Anweisungsnamen durch deren Abkidrzungen, setzt die Zeilennummer um und Uberpruft die
syntaktische Richtigkeit. Er kennt auller NEW die folgenden Kommandos:

LIST {zeilennummer}

RUN

CONT {zeilennummer}

STEP {zeilennummer}

Mit LIST wird das gesamte Programm bzw. die angegebene Zeile aufgelistet. Dricken von Return
nach LIST mit Zeilennummer fuhrt zum Auflisten der Folgezeile. RUN startet das Programm an dessen
Anfang, wahrend man mit CONT von der eingegebenen Zeilennummer ab starten kann. Daruber
hinaus kann mit diesem Kommando die Programmabarbeitung nach STOP fortgesetzt werden. STEP
schlieBlich gestattet das Abarbeiten einer einzelnen Zeile. AnschlieBendes Driicken von Return
bewirkt das Abarbeiten der Folgezeile.

Anhang B. Syntaxbeschreibung von Tiny MPBASIC

Metalinguistische Konstanten sind mit GroBbuchstaben gedruckt. Das sind Zeichen bzw.
Zeichenketten, die unverandert in die konkrete Anweisung ibernommen werden. Die
metalinguistischen Variablen sind klein gedruckt. Sie werden in der konkreten Anweisung durch
andere Zeichen bzw. Zeichenketten gemal§ der Syntaxdefinitionen ersetzt (vgl. Erlauterungen auf S.
17). Weiterhin bedeutet:

-> verbale Erlauterungen

=> Definitionszeichen

| oder

{xxx} darf weggelassen werden

{xxx}* darf beliebig oft dastehen oder auch weggelassen werden
{xxx}*n darf maximal n mal dastehen oder auch weggelassen werden

Syntaxbeschreibung

programm => zeile {zeile}*

zeile => zeilennummer anweisung {; anweisung}*

zeilennummer => pkonst

anweisung => anweisungsname anweisungsargumente {,

anweisungsargumente}

anweisungsname => LET|PROC|GOTO|IF[ELSE|RETURN|GOSUB[WAIT |REM|
CALL |STOP|END | TRAP|CLRTRP|PRINT| PRINTHEX | INPUT

anweisung => (s. Tafel 10, S. 64)

ausdruck => aausdruck/lausdruck

aausdruck => wert {alop wert}*

wert => konst|var|fkt| (aausdruck)

alop => +|-|*|/|$MOD|$AND[$OR | $XOR

konst => pkonst | nkonst

var => buchst

fkt => funktionsname {parameterliste}

pkonst -> ganze Zahl G mit 0<=G<=32767
nkonst => -pconst

funktionsname => name

name => buchst buchst|ziffer {buchst|ziffer}*252
buchst => A|B|C ... X]|Y|Z

ziffer => 1|2|3 ... 8|9|0

lausdruck  => aausdruck vop aausdruck

vop => >|=|<|>=|<=|<>

variablenliste => {var{, var}*}

prozedurname => name

parameterliste => [aausdruck {, aausdruck}*]
text -> bei. ASCII-Zeichenfolge ohne "
kommentar -> bei. ASCII-Zeichenfolge ohne ; und ohne Return

